Программирование под ATM-Turbo
Программная модель ATM2 поддержана в компьютерах ATM-Turbo 2 от ATM (1992), Turbo2 и Turbo2+ от МикроАрт (1992, 1993), Turbo2+ от NedoPC (2004), Pentagon 2.666 и 2.666LE от King of Evil (2008, 2009), ZX Evolution от NedoPC (2009), ATM3 от Zorel’а (2017) и во множестве эмуляторов, из которых для разработчика самый важный на данный момент – UnrealSpeccy.
В ATM-Turbo 2 и Turbo2 было 512K, а в Turbo2+ 1 мегабайт ОЗУ. Конечно, при наличии кусачек, спирта и огурца можно припаять мегабайт и к ATM-Turbo 2 – известно о поддержке мегабайта даже на ATM-Turbo 1 (1991), но там всё другое, кроме экранов и звука, и этих машин на руках уже нет.
Начиная с Pentagon 2.666, в этих компьютерах поддерживается также программная модель ATM3, которая отличается наличием 4096 цветов (а не 64) в палитре и ОЗУ до 4 мегабайт (на Pentagon 2.666(LE) 2 мегабайта, из которых несколько страниц зарезервированы под ПЗУ).
Рекомендую установить Total Commander и Notepad++ с подсветкой исходников *.asm и кнопкой F9 для вызова build.bat в текущем каталоге. Такой Notepad++ лежит у нас в Дискорд-канале. В репозитории NedoOS (http://nedoos.ru/) среди всего прочего есть настроенный UnrealSpeccy и каталог tools с разными полезными утилитами.
Программы под ATM-Turbo можно писать под TR-DOS, CP/M, TASiS (специальная ATM-овская версия iS-DOS) и NedoOS. Поскольку программы под NedoOS тоже в случае чего можно выпустить на дискете TR-DOS, то лучше сразу писать под NedoOS, разве что если это что-то экзотическое или если очень важно время загрузки с реальной дискеты.
Отдельный случай – если вы пишете игру на Си с использованием старого Evo SDK, тогда лучше воспользоваться готовыми исходниками игр в качестве примера. Evo SDK есть и под NedoOS, там его можно расширять любыми ассемблерными функциями (например, прокруткой экрана по X и Y – см. там примеры игр в исходниках).
Память
ATM2 и ATM3 дают переключать страницы по 16K в четырёх окнах адресного пространства: 0x0000, 0x4000, 0x8000, 0xC000. Порты ATM2 для этого соответственно 0x3ff7, 0x7ff7, 0xbff7, 0xfff7 (скрытые), туда надо писать значение 0x7f–page, всего доступно 64 страницы ОЗУ (значения с D6=0 включают ПЗУ, а значения с D7=0 включают управление данным окном памяти через порт 0x7ffd). На ATM3 памяти уже больше, поэтому кроме этих портов есть дублирующие (0x37f7, 0x77f7, 0xb7f7, 0xf7f7), которые переключают 256 страниц ОЗУ (туда надо писать уже 0xff–page). 
Чтобы обо всём этом не думать, NedoOS предоставляет универсальные вызовы (макросы): OS_SETMAINPAGE, SETPG4000, SETPG8000, SETPGC000. Последние три сводятся к call...ld bc,port:out (c),a:ret, это просто и быстро. А вот с окном 0x0000 сложнее, потому что там лежит керналь ОС. Макрос OS_SETMAINPAGE включает страницу, указанную в регистре E. Но вызов внутри этого макроса будет возвращаться через этот же керналь (а ещё он запортит все регистры), и если его не будет в этой странице, то программа вылетит в открытый космос. Считайте, что адреса 0x0100-0xffff полностью ваши, а то, что ниже, либо не переключайте, либо скопируйте из запущенной программы. Обычно в 0x0100-0x3fff  лежит основная часть программы и стек, их переключать не надо, а дальше уже – переключаемые данные и экраны.
Если вы готовы пожертвовать многозадачностью, то можете использовать макрос OS_GETMEMPORTS. Вы получите номера портов в регистрах: ix=memport0000, bc=memport4000, de=memport8000, hl=memportc000, дальше пишите в эти порты что угодно под свою ответственность, только не забудьте перехватить прерывание (по умолчанию оно IM 1).
А что писать в порты по уму? А писать туда надо номера страниц (значения для вывода в порт), которые вам дала система:
- те 4 страницы, которые вам дали при запуске программы: макрос OS_GETMAINPAGES возвращает их в регистрах d,e,h,l=pages in 0000,4000,8000,c000;
- заказывайте себе ещё страницы через OS_NEWPAGE (на выходе a=0 (OK)/!=0 (fail), e=page);
- можете освобождать ненужные страницы через OS_DELPAGE (на входе укажите e=page). Это обычно используется в системных программах. В любом случае, на выходе из вашей программы (макрос QUIT или ручной kill из другой программы) ось сама всё подчистит, а также закроет ваши файлы и сетевые сокеты;
- текущие страницы экрана берите из ячеек (user_scr0_low), (user_scr0_high), (user_scr1_low), (user_scr1_high). Берите непосредственно перед использованием, потому что вашу программу могли свернуть, и номера будут не от физического экрана, а неизвестно откуда. Не пытайтесь выдумывать номера страниц, потому что в адресациях ATM2 и ATM3 номера страниц экрана (то есть значения, выводимые в порт) разные. Хотя и там, и там экраны лежат в страницах 1,5 (scr0) и 3,7 (scr1).
Если вашей программе нужна сплошная нумерация страниц, то рекомендуется сделать так:
align 256
tpgs

ds 256

При старте заполните эту табличку значениями из OS_NEWPAGE (только не жадничайте, не пытайтесь запросить все 256 страниц, вам их не дадут), а страницу перед включением пересчитывайте так:

ld b,tpgs/256
ld c,a
ld a,(bc)
Графика
Все ATMы умеют показывать стандартный экран «6912», а также 320x200 (16 цветов на точку) и 640x200 (аппаратный мультиколор). Начиная с ATM2, появился текстовый режим 80x25, но в NedoOS лучше использовать стандартный терминал через stdio.asm (см. пример hello). Тогда ваша текстовая программа может работать с повышенным числом строк, с сохранением прокрученного текста или даже через telnet.
Наиболее интересен для игроделов режим 320x200, остановимся на нём подробнее.
Имеется, как обычно, два экранных буфера - scr0 и scr1. (Рекомендую использовать оба, если хотите возить спрайты по экрану: в одном рисуете, другой в это время показывается.)
Страница (user_scrN_low):
+0x0000...0x1F3F - pixel pairs 0,4,8, …148,152,156 (8000 bytes).
+0x2000...0x3F3F - pixel pairs 2,6,10, …150,154,158 (8000 bytes).
Страница (user_scrN_high):
+0x0000...0x1F3F - pixel pairs 1,5,9, …149,153,157 (8000 bytes).
+0x2000...0x3F3F - pixel pairs 3,7,11, …151,155,159 (8000 bytes).
В каждом байте по два пикселя:
D6,D2,D1,D0 - left pixel colour (0...15).
D7,D5,D4,D3 - right pixel colour (0...15).
Почему такой странный порядок? Потому что это было технически реализовано как вывод атрибутов. И разбиение на 4 части оттуда же – тикает тот же счётчик, что и в режиме «6912». То же самое и в девайсе «16col» на одной микросхеме для Пентагона.
Поэтому обычно блоки графики выводят по столбцам. Например (экран в 0x4000,0x8000):
primgega_pixsz
;b=hgt,c=wid
;de=gfx
;hl=scr
primgega0
        push bc
        ld hx,b
        push hl
        ld bc,40
primgegacolumn0
        ld a,(de)
        inc de
        ld (hl),a
        add hl,bc
        dec hx
        jr nz,primgegacolumn0
        pop hl
        ld a,0x9f;0xa0
        cp h
        ld bc,0x4000
        adc hl,bc
        jp pe,primgegacolumn0q ;в половине случаев
;8000->с000 (надо 6000) или a000->e001 (надо 4001)
         inc a
        xor h
        ld h,a
primgegacolumn0q
        pop bc
        dec c
        jr nz,primgega0
        ret
Процедуру вывода спрайтов лучше взять из sprexamp (файл prspr.asm). Она умеет клипировать со всех сторон под нужный размер активной зоны. Там же есть модули для вывода фона со скроллом (вертикальным или любым, с тайлами и без).
Графику картинок и спрайтов, а также палитры (которые устанавливаются через OS_SETPAL, адрес палитры в de) проще всего конвертировать утилитой nedores.exe по описаниям в текстовых файлах. Там поддержано много форматов. См. примеры loadscr (проще), br (сложнее). Можно брать целые блоки спрайтов с автонумерацией их меток. Эа утилита выгружает результат в формате ассемблера (*.ast – временный файл ассемблерного исходника). Вы можете использовать эти метки непосредственно или составить таблицу ссылок на них. В конце спрайта лежит dw prsprqwid – адрес выхода из печати спрайта. Поскольку вы же это и компилируете, можете указать любой удобный.
Если нужны картинки во весь экран, то их лучше конвертировать в формат *.16c – это копия экранной памяти (32768 байт) + палитра (32 байта). Для конверсии из *.bmp в *.16c есть утилита 4bbpTo16c.exe, см. пример noise.
А как же включить нужный видеорежим? Например, так (цитирую начало программы):
        DEVICE ZXSPECTRUM1024
        include "../_sdk/sys_h.asm"
        org PROGSTART
begin
        ld e,0+0x80 ;EGA+keep
        OS_SETGFX ;e=0:EGA, e=2:MC, e=3:6912, e=6:text ;+SET FOCUS ;e=-1: disable gfx (out: e=old gfxmode)
Если вы не вызовете OS_SETGFX, то вам не предоставят никакие экранные страницы и даже фокус. «keep» – это автоматическое сохранение-восстановление экранов при переключении фокуса.
В конце исходника обычно ставится:
end

savebin "myprog.com",begin,end-begin

LABELSLIST "../../us/user.l",1 ;это чтобы видеть метки в отладчике UnrealSpeccy
Звук
Музыку на AY можно играть плейером ptsplay.asm. Лучше всего повесить его на системный обработчик звука макросом OS_SETMUSIC (задайте hl=muzaddr (0x4000..0x7fff), a=muzpg, а страницы в 0x8000, 0xc000 берутся из тех, которые воткнуты при вызове). Такая музыка не будет тормозить при работе с другими программами или диском (кроме дискет), в отличие от ручного вызова в вашем обработчике прерывания. Потому что ваш обработчик прерывания вызывается только когда прерывание пришло во время именно вашей задачи.
Цифровые звуки на COVOX тоже лучше играть не руками (вы же не знаете текущую скорость компьютера), а через макрос OS_PLAYCOVOX. Ему передавайте в регистрах: hl=data (0xc000+, байт 0x00=конец), de=pagetable (лежит в 0x0000+), hx=delay (18=11kHz, 7=22kHz, 1=44kHz). Благодаря таблице страниц pagetable можно играть звуки хоть во всю память. Evo SDK в NedoOS работает именно так.
Обработчик прерывания
Вы можете установить свой обработчик при условии, что внутри него вызываете системный. Установка/снятие своего обработчика (который on_int) делается так:
swapimer
        di
        ld de,0x0038
        ld hl,oldimer
        ld bc,3
swapimer0
        ld a,(de)
        ldi ;[oldimer] -> [0x0038]
        dec hl
        ld (hl),a ;[0x0038] -> [oldimer]
        inc hl
        jp pe,swapimer0
        ei
        ret
oldimer
        jp on_int ;заменится на код из 0x0038
        jp 0x0038+3
Лучше возьмите целиком модуль int.asm из примера cyrus2, а дальше можно исправлять. Он уже умеет восстанавливать содержимое стека значением из de (важно для выводилки спрайтов). В sprexamp примерно такой же обработчик, с вариациями.
Работа с диском
Тут всё просто: в NedoOS реализованы стандартные функции OS_OPENHANDLE (возвращает b=filehandle), OS_CLOSEHANDLE, OS_READHANDLE, OS_WRITEHANDLE, а также OS_CREATEHANDLE. Читать/писать можно сколько угодно байт с какого угодно адреса. Система вернёт, сколько реально прочиталось. См. sys_h.asm и file.asm.
Рекомендуется все ресурсы вашей программы положить в подпапочку с тем же именем, что и имя её *.com-файла. При сборке программы все ресурсы скопируются в release. Возьмите build.bat, например, из примера vic20.
Лучше всего программу уже на раннем этапе разработки хранить в репозитории NedoOS. Тогда все возможные изменения в NedoOS коснутся и её. А может, и помощники набегут :)
* * *
Если возникнут проблемы, пишите на dmitry.alonecoder@gmail.com или в Дискорд.
Быстров Дмитрий Михайлович (Alone Coder), 11.08.2022
